Stack Exchange
The root of the trinomial equation z^n+2*z-t written in terms of FoxH
From documentation, the roots of the general trinomial: $$z^n-z-t=0$$ are written in terms of  FoxH special function;
 roots = Exp[(2 \[Pi] I)/(n - 1)]^-j + t/(n - 1)*
 FoxH[{{{0, 1}, {0, n/(n - 1)}}, {}}, {{{0, 1}}, {{-1, 1}, {0, 1/(
    n - 1)}}}, t Exp[(2 \[Pi] I)/(n - 1)]^j];

 n = 6; t = N[1/4];

 z^n - z - t /. z -> Table[roots, {j, 0, n - 2}] // Chop

 (* {0, 0, 0, 0, 0} *)

I tried manipulate formula to solve equation:$$z^n+2*z-t=0$$,but a failed, after few days.
Could anyone help me?
Thanks for your attention.

What about the substitution $z'=-2 z$ and $t'=(-2)^n t$?

@yarchik. No way: Eliminate[{z^n + 2 z - t == 0, 2 z == -x}, z] ?

Part 1. This is an extension of the comment of @yarchik. Set
p[z_,t_]:=z^n-z-t;
q[z_,t_]:=z^n+2*z-t;

I assume that $n>1$ is an integer. Define $b_n \neq 0$ by
b[n_]:=2^(n/(n-1))*Exp[I*Pi*n/(n-1)];

Then
$$q(-b_n z/2, b_n t) = b_n p(z,t)$$
as one can check using something like
Table[q[-b[n]*z/2,b[n]*t]-b[n]*p[z,t],{n,2,20}]//Simplify

Therefore, if $(z,t)$ is a root of $p$ then $(-b_nz/2,b_nt)$ is a root of $q$ and conversely, which we will use in Part 2.
Part 2. OP has given the following formula for roots of $p$:
rootsp[t_,n_]:=Table[Exp[(2 \[Pi] I)/(n-1)]^-j+t/(n-1)*FoxH[{{{0,1},{0,n/(n-1)}},{}},{{{0,1}},{{-1,1},{0,1/(n-1)}}},t Exp[(2 \[Pi] I)/(n-1)]^j],{j,0,n-2}]

By part 1, we then get roots of $q$ using
rootsq[t_,n_]:=-b[n]/2*rootsp[t/b[n],n];

Let us check this in an example:
Block[{n=6,t=N[1/4]},
   q[z,t]/.Map[{z->#}&,rootsq[t,n]]]//Chop
(* {0,0,0,0,0} *)

Also, unless I have misunderstood the question, rootsq is in the form that OP requested, for example rootsq[t,3] gives


I want general formula terms of FoxH solving z^n+2*z-t;=0 ?

Amazing answer 1+).Thanks.
